
G04 – Analysis of Variance

G04BCF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

G04BCF computes the analysis of variance for a general row and column design together with the
treatment means and standard errors.

2 Specification

SUBROUTINE G04BCF(NREP, NROW, NCOL, Y, NT, IT, GMEAN, TMEAN,
1 TABLE, LDT, C, LDC, IREP, RPMEAN, RMEAN, CMEAN,
2 R, EF, TOL, IRDF, WK, IFAIL)
INTEGER NREP, NROW, NCOL, NT, IT(*), LDT, LDC, IREP(NT),
1 IRDF, IFAIL
real Y(NREP*NROW*NCOL), GMEAN, TMEAN(NT),
1 TABLE(LDT,5), C(LDC,NT), RPMEAN(NREP),
2 RMEAN(NREP*NROW), CMEAN(NREP*NCOL),
3 R(NREP*NROW*NCOL), EF(NT), TOL, WK(3*NT)

3 Description

In a row and column design the experimental material can be characterized by a two-way classification,
nominally called rows and columns. Each experimental unit can be considered as being located in a
particular row and column. It is assumed that all rows are of the same length and all columns are of
the same length. Sets of equal numbers of rows/columns can be grouped together to form replicates,
sometimes known as squares or rectangles, as appropriate.

If for a replicate, the number of rows, the number of columns and the number of treatments are equal
and every treatment occurs once in each row and each column then the design is a Latin square. If this
is not the case the treatments will be non-orthogonal to rows and columns. For example in the case of a
lattice square each treatment occurs only once in each square.

For a row and column design, with t treatments in r rows and c columns and b replicates or squares with
n = brc observations the linear model is:

yijk(l) = µ + βi + ρj + γk + τl + eijk

i = 1, 2 . . . , b; j = 1, 2, . . . , r; k = 1, 2 . . . , c; l = 1, 2, . . . , t, where βi is the effect of the ith replicate, ρj

is the effect of the jth row, γk is the effect of the kth column and the ijk(l) notation indicates that the
lth treatment is applied to the unit in row j, column k of replicate i.

To compute the analysis of variance for a row and column design the mean is computed and subtracted
from the observations to give, y′

ijk(l) = yijk(l) − µ̂. Since the replicates, rows and columns are orthogonal
the estimated effects, ignoring treatment effects, β̂i, ρ̂j, γ̂k, can be computed using the appropriate means
of the y′

ijk(l), and the unadjusted sum of squares computed as the appropriate sum of squared totals for
the y′

ijk(l) divided by number of units per total. The observations adjusted for replicates, rows and
columns can then be computed by subtracting the estimated effects from y′

ijk(l) to give y′′
ijk(l).

In the case of a Latin square design the treatments are orthogonal to replicates, rows and columns and so
the treatment effects, τ̂l, can be estimated as the treatment means of the adjusted observations, y′′

ijk(l).
The treatment sum of squares is computed as the sum of squared treatment totals of the y′′

ij(l) divided
by the number of times each treatment is replicated. Finally the residuals, and hence the residual sum
of squares, are given by, rij(l) = y′′

ij(l) − τ̂l.

For a design which is not orthogonal, for example a lattice square or an incomplete latin square, the
treatment effects adjusted for replicates, rows and columns need to be computed. The adjusted treatment

effects are found as the solution to the equations:

Aτ̂ = (R − NbN
T
b /(rc) − NrN

T
r /(bc) − NcN

T
c /(br))τ̂ = q

where q is the vector of the treatment totals of the observations adjusted for replicates, rows and columns,
y′′

ijk(l), R is a diagonal matrix with Rll equal to the number of times the lth treatment is replicated, and
Nb is the t by b incidence matrix, with Nl,i equal to the number of times treatment l occurs in replicate
i, with Nr and Nc being similarly defined for rows and columns. The solution to the equations can be
written as:

τ̂ = Ωq

where, Ω is a generalized inverse of A. The solution is found from the eigenvalue decomposition of A. The
residuals are first calculated by subtracting the estimated adjusted treatment effects from the adjusted
observations to give r′ij(l) = y′′

ij(l) − τ̂l. However, since only the unadjusted replicate, row and column
effects have been removed and they are not orthogonal to treatments, the replicate, row and column
means of the r′ij(l) have to be subtracted to give the correct residuals, rij(l) and residual sum of squares.

Given the sums of squares, the mean squares are computed as the sums of squares divided by the degrees
of freedom. The degrees of freedom for the unadjusted replicates, rows and columns are b − 1, r − 1 and
c − 1 respectively and for the Latin square designs the degrees of freedom for the treatments is t − 1. In
the general case the degrees of freedom for treatments is the rank of the matrix Ω. The F -statistic given
by the ratio of the treatment mean square to the residual mean square tests the hypothesis:

H0 : τ1 = τ2 = . . . = τt = 0.

The standard errors for the difference in treatment effects, or treatment means, for Latin square designs,
are given by:

se(τ̂j − τ̂j∗) =
√

2s2/(bt)

where s2 is the residual mean square. In the general case the variances of the treatment effects are given
by:

Var(τ̂ ) = Ωs2

from which the appropriate standard errors of the difference between treatment effects or the difference
between adjusted means can be calculated.

The analysis of a row-column design can be considered as consisting of different strata: the replicate
stratum, the rows within replicate and the columns within replicate strata and the units stratum. In the
Latin square design all the information on the treatment effects is given at the units stratum. In other
designs there may be a loss of information due to the non-orthogonality of treatments and replicates,
rows and columns and information on treatments may be available in higher strata. The efficiency of
the estimation at the units stratum is given by the (canonical) efficiency factors, these are the non-zero
eigenvalues of the matrix, A, divided by the number of replicates in the case of equal replication, or
by the mean of the number of replicates in the unequally replicated case, see John [3]. If more than
one eigenvalue is zero then the design is said to be disconnected and information on some treatment
comparisons can only be obtained from higher strata.

4 References

[1] Cochran W G and Cox G M (1957) Experimental Designs Wiley

[2] Davis O L (1978) The Design and Analysis of Industrial Experiments Longman

[3] John J A (1987) Cyclic Designs Chapman and Hall

[4] John J A and Quenouille M H (1977) Experiments: Design and Analysis Griffin

[5] Searle S R (1971) Linear Models Wiley

5 Parameters

1: NREP — INTEGER Input

On entry: the number of replicates, b.

Constraint: NREP ≥ 1.

2: NROW — INTEGER Input

On entry: the number of rows per replicate, r.

Constraint: NROW ≥ 2.

3: NCOL — INTEGER Input

On entry: the number of columns per replicate, c.

Constraint: NCOL ≥ 2.

4: Y(NREP*NROW*NCOL) — real array Input

On entry: the n = brc observations ordered by columns within rows within replicates. That is
Y(rc(i − 1) + r(j − 1) + k) contains the observation from the k column of the jth row of the ith
replicate for i = 1, 2, . . . , b; j = 1, 2 . . . , r; k = 1, 2, . . . , c.

5: NT — INTEGER Input

On entry: the number of treatments. If only replicates, rows and columns are required in the
analysis then set NT = 1.

Constraint: NT ≥ 1.

6: IT(*) — INTEGER array Input

Note: the dimension of the array IT must be at least NREP*NROW*NCOL if NT > 1, and 1
otherwise.

On entry: if NT > 1, T(i) indicates which of the NT treatments unit i received, for i = 1, 2, . . . , n.
If NT = 1, IT is not referenced.

Constraint: if NT ≥ 2, 1 ≤ IT(i) ≤ NT, for i = 1, 2, . . . , n.

7: GMEAN — real Output

On exit: the grand mean, µ̂.

8: TMEAN(NT) — real array Output

On exit: if NT ≥ 2, TMEAN(l) contains the (adjusted) mean for the lth treatment, µ̂∗ + τ̂l, for
l = 1, 2, . . . , t, where µ̂∗ is the mean of the treatment adjusted observations yijk(l) − τ̂l. Otherwise
TMEAN is not referenced.

9: TABLE(LDT,5) — real array Output

On exit: the analysis of variance table. Column 1 contains the degrees of freedom, column 2 the
sum of squares, and where appropriate, column 3 the mean squares, column 4 the F -statistic and
column 5 the significance level of the F -statistic. Row 1 is for replicates, row 2 for rows, row 3 for
columns, row 4 for treatments (if NT > 1), row 5 for residual and row 6 for total. Mean squares
are computed for all but the total row, F -statistics and significance are computed for treatments,
replicates, rows and columns. Any unfilled cells are set to zero.

10: LDT — INTEGER Input

On entry: the first dimension of the array TABLE as declared in the (sub)program from which
G04BCF is called.

Constraint: LDT ≥ 6.

11: C(LDC,NT) — real array Output

On exit: the upper triangular part of C contains the variance-covariance matrix of the treatment
effects, the strictly lower triangular part contains the standard errors of the difference between two
treatment effects (means), i.e., C(i, j) contains the covariance of treatment i and j if j ≥ i and the
standard error of the difference between treatment i and j if j < i for i = 1, 2, . . . , t; j = 1, 2, . . . , t.

12: LDC — INTEGER Input

On entry: the first dimension of the array C as declared in the (sub)program from which G04BCF
is called.

Constraint: LDC ≥ NT.

13: IREP(NT) — INTEGER array Output

On exit: if NT > 1, the treatment replications, Rll, for l = 1, 2, . . . , NT. Otherwise IREP is not
referenced.

14: RPMEAN(NREP) — real array Output

On exit: if NREP > 1, RPMEAN(i) contains the mean for the ith replicate, µ̂+β̂i for i = 1, 2, . . . , b.
Otherwise RPMEAN is not referenced.

15: RMEAN(NREP*NROW) — real array Output

On exit: RMEAN(j) contains the mean for the jth row, µ̂ + ρ̂i for j = 1, 2, . . . , r.

16: CMEAN(NREP*NCOL) — real array Output

On exit: CMEAN(k) contains the mean for the kth column, µ̂ + γ̂k for k = 1, 2, . . . , c.

17: R(NREP*NROW*NCOL) — real array Output

On exit: the residuals, ri for i = 1, 2, . . . , n.

18: EF(NT) — real array Output

On exit: if NT ≥ 2, the canonical efficiency factors. Otherwise EF is not referenced.

19: TOL — real Input

On entry: the tolerance value used to check for zero eigenvalues of the matrix Ω. If TOL = 0.0 a
default value of 0.00001 is used.

Constraint: TOL ≥ 0.0

20: IRDF — INTEGER Input

On entry: an adjustment to the degrees of freedom for the residual and total. If IRDF ≥ 1 the
degrees of freedom for the total is set to n − IRDF and the residual degrees of freedom adjusted
accordingly. If IRDF = 0, the total degrees of freedom for the total is set to n − 1, as usual.

Constraint: IRDF ≥ 0.

21: WK(3*NT) — real array Workspace

22: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should
refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL �= 0 on exit,
users are recommended to set IFAIL to −1 before entry. It is then essential to test the value of
IFAIL on exit.

6 Errors and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL= 1

On entry, NREP < 1,

or NROW < 2,

or NCOL < 2,

or NT < 1,

or LDT < 6,

or LDC < NT,

or TOL < 0.0,

or IRDF < 0.

IFAIL=2

On entry, IT(i) < 1 or IT(i) > NT for some i when NT ≥ 2,

or no value of IT = j for some j = 1, 2, . . . , NT, when NT ≥ 2.

IFAIL= 3

On entry, the values of Y are constant.

IFAIL= 4

A computed standard error is zero due to rounding errors, or the eigenvalue computation failed to
converge. Both are unlikely error exits.

IFAIL=5

The treatments are totally confounded with replicates, rows and columns, so the treatment sum
of squares and degrees of freedom are zero. The analysis of variance table is not computed, except
for replicate, row, column and total sums of squares and degrees of freedom.

IFAIL=6

The residual degrees of freedom or the residual sum of squares are zero, columns 3, 4 and 5 of the
analysis of variance table will not be computed and the matrix of standard errors and covariances,
C, will not be scaled by s or s2.

IFAIL=7

The design is disconnected, the standard errors may not be valid. The design may have a nested
structure.

7 Accuracy

The algorithm used in this routine, described in Section 3, achieves greater accuracy than the traditional
algorithms based on the subtraction of sums of squares.

8 Further Comments

To estimate missing values the Healy and Westmacott procedure or its derivatives may be used, see [4].
This is an iterative procedure in which estimates of the missing values are adjusted by subtracting the
corresponding values of the residuals. The new estimates are then used in the analysis of variance. This
process is repeated until convergence. A suitable initial value may be the grand mean. When using this
procedure IRDF should be set to the number of missing values plus one to obtain the correct degrees of
freedom for the residual sum of squares.

For analysis of covariance the residuals are obtained from an analysis of variance of both the response
variable and the covariates. The residuals from the response variable are then regressed on the residuals
from the covariates using, say, G02CBF or G02DAF. The results from those routines can be used to test
for the significance of the covariates. To test the significance of the treatment effects after fitting the
covariate, the residual sum of squares from the regression should be compared with the residual sum of
squares obtained from the equivalent regression but using the residuals from fitting replicates, rows and
columns only.

9 Example

The data for a 5× 5 Latin square is input and the ANOVA and treatment means computed and printed.
Since the design is orthogonal only one standard error need be printed

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* G04BCF Example Program Text
* Mark 17 Release. NAG Copyright 1995.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, NTMAX, NBMAX
PARAMETER (NMAX=25,NTMAX=5,NBMAX=5)

* .. Local Scalars ..
real GMEAN
INTEGER I, IFAIL, J, N, NCOL, NREP, NROW, NT

* .. Local Arrays ..
real C(NTMAX,NTMAX), CMEAN(NBMAX), EF(NTMAX), R(NMAX),

+ RMEAN(NBMAX), RPMEAN(NBMAX), TABLE(6,5),
+ TMEAN(NTMAX), WK(3*NTMAX), Y(NMAX)
INTEGER IREP(NTMAX), IT(NMAX)

* .. External Subroutines ..
EXTERNAL G04BCF

* .. Executable Statements ..
WRITE (NOUT,*) ’G04BCF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) NREP, NROW, NCOL, NT
IF (NROW.LE.NBMAX .AND. NCOL.LE.NBMAX .AND. NT.LE.NTMAX) THEN

N = NREP*NROW*NCOL
READ (NIN,*) (Y(I),I=1,N)
READ (NIN,*) (IT(I),I=1,N)
IFAIL = -1

*
CALL G04BCF(NREP,NROW,NCOL,Y,NT,IT,GMEAN,TMEAN,TABLE,6,C,NTMAX,

+ IREP,RPMEAN,RMEAN,CMEAN,R,EF,0.00001e0,0,WK,IFAIL)
*

WRITE (NOUT,*)
WRITE (NOUT,*) ’ ANOVA TABLE’
WRITE (NOUT,*)
IF (NREP.GT.1) THEN

WRITE (NOUT,99998) ’ Reps ’, (TABLE(1,J),J=1,5)
END IF
WRITE (NOUT,99998) ’ Rows ’, (TABLE(2,J),J=1,5)
WRITE (NOUT,99998) ’ Columns ’, (TABLE(3,J),J=1,5)
WRITE (NOUT,99998) ’ Treatments ’, (TABLE(4,J),J=1,5)

WRITE (NOUT,99998) ’ Residual ’, (TABLE(5,J),J=1,3)
WRITE (NOUT,99998) ’ Total ’, (TABLE(NOUT,J),J=1,2)
WRITE (NOUT,*)
WRITE (NOUT,*) ’ Treatment means’
WRITE (NOUT,*)
WRITE (NOUT,99999) (TMEAN(I),I=1,NT)
WRITE (NOUT,*)
WRITE (NOUT,99997) ’ S.E. of difference (orthogonal design) = ’

+ , C(2,1)
END IF
STOP

*
99999 FORMAT (8F10.4)
99998 FORMAT (A,F3.0,2X,3(F10.4,2X),F8.4)
99997 FORMAT (A,F10.4)

END

9.2 Program Data

G04BCF Example Program Data

1 5 5 5

6.67 7.15 8.29 8.95 9.62
5.40 4.77 5.40 7.54 6.93
7.32 8.53 8.50 9.99 9.68
4.92 5.00 7.29 7.85 7.08
4.88 6.16 7.83 5.38 8.51

5 4 1 3 2
2 5 4 1 3
3 2 5 4 1
1 3 2 5 4
4 1 3 2 5

9.3 Program Results

G04BCF Example Program Results

ANOVA TABLE

Rows 4. 29.4231 7.3558 9.0266 0.0013
Columns 4. 22.9950 5.7487 7.0545 0.0037
Treatments 4. 0.5423 0.1356 0.1664 0.9514
Residual 12. 9.7788 0.8149
Total 24. 62.7392

Treatment means

7.3180 7.2440 7.2060 6.9000 7.2600

S.E. of difference (orthogonal design) = 0.5709

